
Motivation

Distributed consensus in the presence of faults

▶ fundamental problem: servers agree on (a sequence of) proposals

▶ famously unsolvable for asynchronous communication and crash faults (FLP 1985)

▶ existing algorithms ensure safety, rely on sufficient availability for liveness

Raft algorithm

▶ introduced as an “easier to understand” alternative to Paxos

▶ fundamentally designed as an iterated consensus algorithm

▶ every server maintains a sequence (“log”) of entries

▶ logs converge to common prefixes over time

D. Ongaro, J.K. Ousterhout: In search of an understandable consensus algorithm. Usenix ATC 2014.

Stephan Merz (Inria Nancy) Understanding Raft IFIP WG 2.3 2 / 29



Motivation

Distributed consensus in the presence of faults

▶ fundamental problem: servers agree on (a sequence of) proposals

▶ famously unsolvable for asynchronous communication and crash faults (FLP 1985)

▶ existing algorithms ensure safety, rely on sufficient availability for liveness

Raft algorithm

▶ introduced as an “easier to understand” alternative to Paxos

▶ fundamentally designed as an iterated consensus algorithm

▶ every server maintains a sequence (“log”) of entries

▶ logs converge to common prefixes over time

D. Ongaro, J.K. Ousterhout: In search of an understandable consensus algorithm. Usenix ATC 2014.

Stephan Merz (Inria Nancy) Understanding Raft IFIP WG 2.3 2 / 29



Formal Verification of Raft

TLA+ specification written by Ongaro
▶ precise blueprint for implementing the protocol
▶ level of abstraction close to the implementation
▶ validation by simulation and partial proof of some properties
▶ Raft paper and Ongaro’s PhD thesis present a behavioral correctness proof

Mechanized correctness proofs based on inductive invariants
▶ formal proof in Verdi framework: ∼ 50k lines of Rocq proofs
▶ recent proof based on Ongaro’s specification: ∼ 30k lines of TLA+ proofs
▶ heroic achievements, but difficult to understand and extend

Objective: model and verify Raft at a higher level of abstraction

D. Woos et al.: Planning for change in a formal verification of the Raft consensus protocol. CPP 2015.
O. Marcu: announcement on LinkedIn, 2026.

Overview of Raft

Servers and their logs
▶ every server maintains a sequence (“log”) of entries containing proposed values
▶ a prefix of the log is committed (stable), the rest may be subject to modification

Leaders and followers

▶ Raft operates over successive terms managed by a leader
▶ leader is responsible for introducing new entries and for committing entries
▶ if a leader is suspected to have failed, a new leader is elected

committed
trm = 1
val = 3

trm = 1
val = 7

trm = 2
val = 0

trm = 4
val = 5

trm = 4
val = 9

trm = 1

Outline

1 Leader Election

2 Updating the Log

3 Committing Entries

4 Effort of Proof

Lifecycle of Server States

follower candidate leader
timeout elected

update term

update term

Elections and terms

▶ leaders are elected for a term

▶ follower may suspect the leader to have failed and become a candidate for a higher term

▶ servers only vote for candidates at strictly higher terms

▶ a candidate is elected when it receives votes from a quorum of servers

▶ whenever a server learns of the existence of a higher term, it becomes a follower

Timeout and Voting in TLA+

A follower or candidate starts a new election and votes for itself

Timeout(s) ∆
= ∧ role[s] ∈ {“follower”, “candidate”}

∧ role′ = [role EXCEPT ![s] = “candidate”]
∧ ∃t ∈ Term : ∧ t > term[s] ∧ term′ = [term EXCEPT ![s] = t]

∧ ballots′ = [ballots EXCEPT ![s] = @ ∪ {⟨s, t⟩}]

A server votes for a candidate at a higher term

Vote(s) ∆
= ∃cdt ∈ Server \ {s} : LET ct ∆

= term[cdt] IN

∧ term[s] < ct
∧ term′ = [term EXCEPT ![s] = ct]
∧ role′ = [role EXCEPT ![s] = “follower”]
∧ ballots′ = [ballots EXCEPT ![s] = @ ∪ {⟨cdt, ct⟩}]

Electing a Leader and Term Update

A candidate is elected leader when it obtains a quorum of votes

ElectLeader(s) ∆
= ∧ role[s] = “candidate”

∧ ∃Q ∈ Quorum : ∀vt ∈ Q : ⟨s, term[s]⟩ ∈ ballots[vt]
∧ role′ = [role EXCEPT ![s] = “leader”]
∧ term′ = term

A server becomes follower when it learns of a server at a higher term

UpdateTerm(s) ∆
= ∃srv ∈ Server :

∧ term[srv] > term[s]
∧ term′ = [term EXCEPT ![s] = term[srv]]
∧ role′ = [role EXCEPT ![s] = “follower”]

Stephan Merz (Inria Nancy) Understanding Raft IFIP WG 2.3 8 / 29



Electing a Leader and Term Update

A candidate is elected leader when it obtains a quorum of votes

ElectLeader(s) ∆
= ∧ role[s] = “candidate”

∧ ∃Q ∈ Quorum : ∀vt ∈ Q : ⟨s, term[s]⟩ ∈ ballots[vt]
∧ role′ = [role EXCEPT ![s] = “leader”]
∧ term′ = term

A server becomes follower when it learns of a server at a higher term

UpdateTerm(s) ∆
= ∃srv ∈ Server :

∧ term[srv] > term[s]
∧ term′ = [term EXCEPT ![s] = term[srv]]
∧ role′ = [role EXCEPT ![s] = “follower”]

Stephan Merz (Inria Nancy) Understanding Raft IFIP WG 2.3 8 / 29



Election Invariant

BackedByQuorum(srv, t) ∆
= ∃Q ∈ Quorum : ∀vt ∈ Q : ⟨srv, t⟩ ∈ ballots[vt]

ElectionInv ∆
= ∧ ∀cdt, srv ∈ Server : ∀t ∈ Term : ⟨cdt, t⟩ ∈ ballots[srv] ⇒

∧ 0 < t ∧ t ≤ term[cdt] ∧ t ≤ term[srv]
∧ ∀b ∈ ballots[srv] : b[2] = t ⇒ b = ⟨cdt, t⟩

∧ ∀srv ∈ Server : role[srv] ∈ {“candidate”, “follower”} ⇒ term[srv] > 0
∧ ∀ldr ∈ Server : role[ldr] = “leader” ⇒ BackedByQuorum(ldr, term[ldr])

a server votes only for one candidate at any given term

the term of any ballot is a lower bound for the terms of the voter and the candidate

leaders are backed by a quorum of voters

since any two quorums intersect, there can only be one leader per term

LEMMA BBQSameTerm ∆
= ASSUME ElectionInv, NEW s1, s2 ∈ Server, NEW t ∈ Term,

BackedByQuorum(s1, t), BackedByQuorum(s2, t)
PROVE s1 = s2

Outline

1 Leader Election

2 Updating the Log

3 Committing Entries

4 Effort of Proof

Servers and their Logs

1 2 3 4 5 6

entries are tagged with the term during which they were added

leaders append entries to their logs

followers copy entries from their leader

Adding an Entry to the Log of a Leader

A leader serves a client request

AppendEntry(s) ∆
= ∧ role[s] = “leader”

∧ ∃v ∈ Value : entries′ = [entries EXCEPT ![s] =
Append(@, [val 7→ v, trm 7→ term[s]])]

the entry is tagged with the leader’s term and appended to the log

trm = 1

trm = 1
val = 7

trm = 2

trm = 2
val = 8 leader, before

trm = 1

trm = 1

trm = 2
val = 5 leader, after

Updating the Log of a Follower

Find the first entry at which the logs differ and copy the leader’s entry

LearnEntry(s) ∆
= ∃ldr ∈ Server :

∧ role[s] = “follower”
∧ BackedByQuorum(ldr, term[s])
∧ term[ldr] = term[s]
∧ ∃n∈ 1 .. Min(Len(entries[s]) + 1, Len(entries[ldr])) :

∧ n − 1 ∈ 1 .. Len(entries[s]) ⇒ entries[s][n − 1].trm = entries[ldr][n − 1].trm
∧ n ∈ 1 .. Len(entries[s]) ⇒ entries[s][n].trm ̸= entries[ldr][n].trm
∧ entries′ = [entries EXCEPT ![s] = Append(SubSeq(entries[s], 1, n − 1), entries[ldr][n])]

trm = 1
val = 3

trm = 1
val = 7

trm = 1
val = 2

trm = 3
val = 0

trm = 4
val = 5

trm = 4
val = 9 leader

trm = 1
val = 3

trm = 1
val = 7

trm = 2
val = 6

trm = 2
val = 8

trm = 2
val = 5 follower, before

trm = 1
val = 3

trm = 1
val = 7

trm = 1
val = 2 follower, after

LearnEntry may
overwrite logs

Log Invariants (1)

Every entry was introduced by a leader for its term

EntryHasCreator ∆
= ∀s ∈ Server : ∀i ∈ 1 .. Len(entries[s]) : ∃c ∈ Server :

∧ BackedByQuorum(c, entries[s][i].trm)

∧ ∨ role[c] = “leader” ∧ term[c] = entries[s][i].trm
∨ term[c] > entries[s][i].trm

A leader for term t holds all entries of term t that appear anywhere

InLog(s, i, e) ∆
= i ∈ 1 .. Len(entries[s]) ∧ entries[s][i] = e

LeaderHasEntries ∆
= ∀ldr, s ∈ Server : ∀i ∈ 1 .. Len(entries[s]) :

role[ldr] = “leader” ∧ entries[s][i].trm = term[ldr] ⇒ InLog(ldr, i, entries[s][i])

Log Invariants (2)

Entry terms are bounded by the term of the server and monotonic

EntryTerm ∆
= ∀s ∈ Server : ∀i ∈ 1 .. Len(entries[s]) : entries[s][i].trm ≤ term[s]

EntryTermMonotonic ∆
= ∀s ∈ Server : ∀i, j ∈ 1 .. Len(entries[s]) :

i ≤ j ⇒ entries[s][i].trm ≤ entries[s][j].trm

The position and term of an entry uniquely determines the prefix of the log

LogMatching ∆
= ∀s1, s2 ∈ Server : ∀i ∈ 1 .. Min(Len(entries[s1], Len(entries[s2]))) :

entries[s1][i].trm = entries[s2][i].trm ⇒ ∀j ∈ 1 .. i : entries[s1][j] = entries[s2][j]

These invariants are inductive relative to the election invariant

Outline

1 Leader Election

2 Updating the Log

3 Committing Entries

4 Effort of Proof

When does an Entry Become Stable?

Eventually, entries should stabilize across the network
▶ idea: commit an entry when it has been replicated on a quorum of servers
▶ Can the entry of term 2 be declared committed?

No!

trm = 1 trm = 2 trm = 4

trm = 1 trm = 2

trm = 1 trm = 2

trm = 1

trm = 1 trm = 3

⇝

trm = 1 trm = 2 trm = 4

trm = 1 trm = 3

trm = 1 trm = 3

trm = 1 trm = 3

trm = 1 trm = 3 trm = 5

Committing entries is delicate

Primary Commits

A leader may commit an entry introduced in its current term . . .
▶ . . . provided that entry is backed by a quorum of followers at that term
▶ the entire prefix of the log is implicitly committed as well

commit

S1 trm = 1
val = 3

trm = 1
val = 7

trm = 1
val = 2

trm = 3
val = 0

trm = 4
val = 5

S2 trm = 1
val = 3

trm = 1

val = 2

trm = 3
val = 0

trm = 4
val = 5

S3 trm = 1
val = 3

trm = 1
val = 7

trm = 1
val = 2

trm = 3
val = 0

trm = 4
val = 5

trm = 4
val = 9

S4 trm = 1
val = 3

trm = 1
val = 7

trm = 2
val = 6

trm = 2
val = 8

S5 trm = 1
val = 3

trm = 1
val = 7

trm = 1
val = 2

Must prevent servers S4 and S5 from being elected leaders!

The Voting Proviso

Add a precondition to the Vote action

lastEntryTerm(es) ∆
= IF es = ⟨⟩ THEN 0 ELSE es[Len(es)].trm

Vote(s) ∆
= ∃cdt ∈ Server \ {s} : LET ct ∆

= term[cdt] IN

∧ term[s] < ct ∧ ∀b ∈ ballots[s] : b[2] < ct
∧ ∨ lastEntryTerm(entries[s]) < lastEntryTerm(entries[cdt])
∨ ∧ lastEntryTerm(entries[s]) = lastEntryTerm(entries[cdt])
∧ Len(entries[s]) ≤ Len(entries[cdt])

∧ term′ = [term EXCEPT ![s] = ct]
∧ role′ = [role EXCEPT ![s] = “follower”]
∧ ballots′ = [ballots EXCEPT ![s] = @ ∪ {⟨cdt, ct⟩}]

Must explain why this is enough, given that logs are not stable!

Leader Commit

A leader commits an entry from the current term backed by a quorum

LeaderCommit(s) ∆
=

∧ role[s] = “leader”
∧ ∃n ∈ (commitIndex[s] + 1) .. Len(entries[s]) :

∧ entries[s][n].trm = term[s]
∧ ∃Q ∈ Quorum : ∧ ∀sq ∈ Q : InLog(sq, n, entries[s][n]) ∧ term[sq] = term[s]

∧ commitQuorum′ = [commitQuorum EXCEPT ![s][n] = Q]

∧ commitIndex′ = [commitIndex EXCEPT ![s] = n]

▶ commitQuorum is a history variable and records the backing quorum for the proof
▶ commitIndex indicates for each server the committed prefix of the log

Adaptation of action LearnEntry
▶ if necessary, adjust commitIndex so that it doesn’t exceed the length of the log

Push Information on Committed Entries to Followers

Followers learn about committed entries from other servers

GossipCommit(s) ∆
= ∃srv ∈ Server : ∃n ∈ (commitIndex[s] + 1) .. commitIndex[srv] :

∧ role[s] = “follower”
∧ term[s] ≤ term[srv]
∧ InLog(s, n, entries[srv][n])
∧ term′ = [term EXCEPT ![s] = term[srv]]
∧ commitIndex′ = [commitIndex EXCEPT ![s] = n]

▶ any entry that is committed somewhere and present in the log can be committed

Invariants on Committed Entries (1)

commitQuorum[s][i] is either empty or contains a quorum, and s is a leader

CQType ∆
= ∀s ∈ Server, i ∈ Index : commitQuorum[s][i] ̸= ∅ ⇒

∧ i ∈ 1 .. commitIndex[s]
∧ commitQuorum[s] ∈ Quorum
∧ BackedByQuorum(s, entries[s][i].trm)

Every committed entry is dominated by a leader commit

PrimaryCommit(ldr, j) ∆
= commitQuorum[ldr][j] ∈ Quorum

CommitHasPrimary ∆
=

∀s ∈ Server : ∀i ∈ 1 .. commitIndex[s] : ∃ldr ∈ Server : ∃j ∈ i .. commitIndex[ldr] :
∧ PrimaryCommit(ldr, j)
∧ InLog(ldr, i, entries[s][i])
∧ term[s] ≥ entries[ldr][j].trm

Invariants on Committed Entries (2)

All members of a commit quorum only vote for servers that contain the entry

CommitQuorum ∆
= ∀ldr ∈ Server : ∀i ∈ 1 .. Len(entries[ldr]) : PrimaryCommit(ldr, i) ⇒

∀s ∈ commitQuorum[ldr][i] : ∧ InLog(s, i, entries[ldr][i]) ∧ term[s] ≥ entries[ldr][i].trm
∧ ∀cdt ∈ Server, t ∈ Term :

t > entries[ldr][i].trm ∧ ⟨cdt, t⟩ ∈ ballots[s] ⇒
InLog(cdt, i, entries[ldr][i])

Servers that are more up to date than a leader commit contain the committed entry

MoreUpToDateHasCommit ∆
= ∀ldr, s ∈ Server : ∀i ∈ 1 .. Len(entries[ldr]) : ∀j ∈ 1 .. Len(entries[s]) :

PrimaryCommit(ldr, i) ∧ ∨ entries[s][j].trm > entries[ldr][i].trm
∨ entries[s][j].trm = entries[ldr][i].trm ∧ j > i

⇒ InLog(s, i, entries[ldr][i])

Invariants focus on primary commits

Leader Completeness

Consequence of the commit invariant

▶ If a server is a leader for a term higher than that of a primary commit,
it must contain the committed entry.

LeaderComplete ∆
= ∀ld1, ld2 ∈ Server : ∀i ∈ 1 .. Len(entries[ld1]) : ∀t ∈ Term :

∧ PrimaryCommit(ld1, i)
∧ BackedByQuorum(ld2, t)
∧ t > entries[ld1][i].trm

⇒ InLog(ld2, i, entries[ld1][i])

Therefore, a primary commit can never be overwritten by LearnEntry

▶ the Raft paper can be read as asserting that property for all committed entries

▶ such an entry may in fact be overwritten and restored later

No Entry in the Committed Prefix is Overwritten

Entries that a server knows to be committed are stable

Proof. Assume s ∈ Server and i ∈ 1 .. commitIndex[s], define ci ∆
= commitIndex[s].

By CommitHasPrimary, pick ldr ∈ Server and j ∈ ci .. commitIndex[ldr] such that:
PrimaryCommit(ldr, j), InLog(ldr, ci, entries[s][ci]), term[s] ≥ entries[ldr][j].trm.
The only action potentially overwriting entries[s][i] is LearnEntry(s), so consider
srv ∈ Server, BackedByQuorum(srv, term[s]), term[srv] = term[s],
n ∈ Min(Len(entries[s]) + 1, Len(entries[srv])) s.t.
n ∈ 1 .. Len(entries[s]) ⇒ entries[s][n].trm ̸= entries[srv][n].trm.
Assume moreover that n ≤ ci (otherwise entries[s][i] is preserved).
It follows that term[srv] ≥ entries[ldr][j].trm. We prove InLog(srv, j, entries[ldr][j]).
▶ If term[srv] > entries[ldr][j].trm, this follows from LeaderComplete.
▶ If term[srv] = entries[ldr][j].trm, then srv = ldr by BBQSameTerm and CQType.

By LogMatching and ci ≤ j, it follows that InLog(srv, ci, entries[s][ci]).

Using again LogMatching and n ≤ ci, we have InLog(srv, n, entries[s][n]), contradiction. QED

Main Safety Property

Any two servers agree on committed entries at the same position.

THEOREM Safety ∆
= ASSUME TypeOK, LogInv, CommitInv,

NEW s1 ∈ Server, NEW s2 ∈ Server,
NEW i ∈ 1 .. Min(commitIndex[s1], commitIndex[s2])

PROVE entries[s1][i] = entries[s2][i]

Proof. Using CommitHasPrimary and LogMatching, pick
ld1 ∈ Server and Q1 ∈ Quorum such that InLog(srv, i, entries[s1][i] for all srv ∈ Q1 and
ld2 ∈ Server and Q2 ∈ Quorum such that InLog(srv, i, entries[s2][i] for all srv ∈ Q2.
Pick some srv ∈ Q1 ∩ Q2 to prove the conclusion. QED

Outline

1 Leader Election

2 Updating the Log

3 Committing Entries

4 Effort of Proof

TLA+ Specifications and Proofs

Size of the development (loc, without comments/blank lines)

definitions lemmas invariant proof
specification 115 67 29
election invariant 11 54 158
log invariant 26 37 430
commit invariant 42 169 674
totals 194 327 1291

Summing Up

High-level specification of Raft
▶ understand the principles on which the algorithm is built
▶ exhibit and prove an inductive invariant
▶ facilitate the study of variations and extensions

Main insights gained
▶ overall correctness argument follows the structure of the original paper
▶ importance of focusing on primary commits in the proof
▶ certain actions (updating terms, secondary commits) need not be driven by leader

Work in progress
▶ refine the specification to better localize some actions
▶ validate Raft implementations against the resulting specification

